
Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iWD-EUTRA-B2011-03_D12wk05’.

Corrections 

Change 1 – Correction to function ‘f_TC_9_2_3_1_17_GERAN()’ 

	Function name
	f_TC_9_2_3_1_17_GERAN()

	Reason for change
	1. At step 11a2 the Routing Area Update Reject message is sent but TTCN had not yet prepared the DL TBF in order to transmit this message. 

2. And immediately after sending the Routing Area Update Reject message (at the TTCN level) the cell power level of cell 24 is changed to “Non-Suitable Cell”. Which means that the power level was changed in the middle of the sending procedure and thus may cause the procedure to fail.

	Summary of change
	1. ‘G_LLC.send’ is replaced by calling function ‘f_GPRS_SendDownlinkMsG()’ with the same parameters, thus preparing DL TBF in order to transmit the RAU Rej message.

2. As in many other situations in the ATS a small delay is introduced to allow the send procedure to complete before actually changing the power level. 

	TTCN module
	9_2\NAS_TrackingArea_UG_GERAN.ttcn

	MCC160 Comment
	


Before change

	  //---------------------------------------------------------------------
  /*
   * @desc      REFERENCE TS 36.523-1 clause 9.2.3.1.17
   * @status    APPROVED
   */
  function f_TC_9_2_3_1_17_GERAN () runs on GERAN_PTC
  { /* Test Purpose (see f_TC_9_2_3_1_17_EUTRA) */
    /* @sic R5-113696: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */
    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;
    timer t_Wait30s := f_UG_SetTimerToleranceMax (30.0);
    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test @sic R5s110318 sic@
    f_GERAN_CreateCell (geran_Cell24);
    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params
    f_GERAN_TestBody_Set( true );
    // First Trigger:
    //@siclog "Step 10" siclog@
    // Set Cell 24 to "Serving Cell"
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);
    //+ Step
    //@siclog "Step 11a1" siclog@
    // Check: Does the UE transmit a ROUTING AREA UPDATE REQUEST message in the next 30 seconds on cell 24?
    t_Wait30s.start;
    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,
                                             car_G_LLC_UnitData_IND (cr_G_RA_UpdReqAny (cr_GMM_UpdateType,
                                                                                        cr_RAI (omit),
                                                                                        cr_PTMSI_SignatureAny ifpresent,
                                                                                        ?)));
    t_Wait30s.stop;
    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 11a1");
    //@siclog "Step 11a2" siclog@
    // Send ROUTING AREA UPDATE REJECT message with cause "Roaming not allowed in this area"
    G_LLC.send(cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),
                                       tsc_LLCSapi_GMM,
                                       tsc_LLC_PM,
                                       px_GSM_CipheringOnOff,
                                       cs_G_RAUpdReject(cs_GMM_Cause('00001101'B)))); // cause 'Roaming not allowed in this area'
    //+ Set the cell type of cell 24 to the "Non-Suitable cell".
    //@siclog "Step 12" siclog@
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);
    // Test case now finished on GERAN side, Finished here for now
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    f_GERAN_TestBody_Set( false );
    // Test case finished, so take cell down
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SS_CellRelease ( geran_Cell24 );
  } // f_TC_9_2_3_1_17_GERAN
  


After change

	  //---------------------------------------------------------------------
  /*
   * @desc      REFERENCE TS 36.523-1 clause 9.2.3.1.17
   * @status    APPROVED
   */
  function f_TC_9_2_3_1_17_GERAN () runs on GERAN_PTC
  { /* Test Purpose (see f_TC_9_2_3_1_17_EUTRA) */
    /* @sic R5-113696: pc_GERAN and px_RATComb_Tested are checked at top-level sic@ */
    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;
    timer t_Wait30s := f_UG_SetTimerToleranceMax (30.0);
    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test @sic R5s110318 sic@
    f_GERAN_CreateCell (geran_Cell24);
    f_GERAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params
    f_GERAN_TestBody_Set( true );
    // First Trigger:
    //@siclog "Step 10" siclog@
    // Set Cell 24 to "Serving Cell"
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);
    //+ Step
    //@siclog "Step 11a1" siclog@
    // Check: Does the UE transmit a ROUTING AREA UPDATE REQUEST message in the next 30 seconds on cell 24?
    t_Wait30s.start;
    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,
                                             car_G_LLC_UnitData_IND (cr_G_RA_UpdReqAny (cr_GMM_UpdateType,
                                                                                        cr_RAI (omit),
                                                                                        cr_PTMSI_SignatureAny ifpresent,
                                                                                        ?)));
    t_Wait30s.stop;
    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 11a1");
    //@siclog "Step 11a2" siclog@
    // Send ROUTING AREA UPDATE REJECT message with cause "Roaming not allowed in this area"
    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),
                                       tsc_LLCSapi_GMM,
                                       tsc_LLC_PM,
                                       px_GSM_CipheringOnOff,
                                       cs_G_RAUpdReject(cs_GMM_Cause('00001101'B)))); // cause 'Roaming not allowed in this area'
    f_Delay (1.0);  //Wait for NAS message to be sent
    //+ Set the cell type of cell 24 to the "Non-Suitable cell".
    //@siclog "Step 12" siclog@
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);
    // Test case now finished on GERAN side, Finished here for now
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    f_GERAN_TestBody_Set( false );
    // Test case finished, so take cell down
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SS_CellRelease ( geran_Cell24 );
  } // f_TC_9_2_3_1_17_GERAN

