
1)  Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 9.2.1.2.1d, which is part of the LTE test suite. 

The test case can be demonstrated to run with an LTE UE (see section 6). Execution log files are provided as evidence. 

3) Verification Test Summary

Test Case:
9.2.1.2.1d
Test Group:
Idle Mode
ATS Version:
iwd-EUTRA-B2011-03_D12wk09
System Simulator used:
Anite Conformance Toolset
UE used:
QualComm MDM, Nvidia ICERA 410
Verification Status:
PASS
4) Corrections required for test case 9.2.1.2.1d

4.1 Introduction

This section describes the changes required to make test case 9.2.1.2.1d run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd- iwd-EUTRA-B2011-03_D12wk09 release.

4.2 Change 1

	Testcase name
	9.2.1.2.1d

	Reason for change
	1- As Combined RAU si performed at the UTRA cell, when UE comes back to LTE cell, UE will transmit OldPtmsiSignature, AdditionalGuti & Nonce. 

2- TP can be achieved using default USIM card.

3- The different vairables are declared to be used in Actiavting the default bearer.

v_PdnConnectivityRequest,v_EIT_Flag,v_EPS_TI,v_Pco,v_APN,v_EPS_ContextStatus

4- At step1 the power of EUTRA cell will be made Serving to allow the UE to camp on the EUTRA cell.

5- At step 11 v_PdnConnectivityRequest,v_EIT_Flag,v_EPS_TI,v_Pco,v_APN,v_EPS_ContextStatus will be computed from the attach request to used in activating the default bearer

6- At step 2-13 acivate the default bearer request will be send piggybacking with the attach accept.

7- At step 22, the UE transmits an RRCConnectionSetupComplete message on Cell 1.This message includes an EXTENDED SERVICE REQUEST message. A valid EPS security context has previously been established during the preamble. Therefore, according to 3GPP TS 24.301 4.4.2.3 2) and 4.4.5, when a UE in EMM-IDLE mode establishes a new NAS signalling connection it shall send the initial NAS message unciphered. 

8- At step 23 the EUTRA cell will be made Non Suitable (As During CS fallback UE can try to access EUTRA).

9- After step 23 the AuthParams will be sent by Eutra cell A (as all the signalling is performed at CELL A).

10- At step 43 EUTRA cell will be made serving to allow UE to perform Tracking area update procedure.

11- At step 43-48 the Tracking area update function will not contain the update type “tsc_EpsUpdate_Combined_IMSIAttach” and in the TAU accept will not be sending the CSnotPreferred in Additional update IE as per the test puprose.

As per 24.301 clause 5.5.3.3.3 point d) the EPS Update type in the TAU request should be set to "combined TA/LA updating with IMSI attach".

12- At step 48 the RRC connection will be released after the TAU procedure.

13- At step 51 the Extended service request will be of MO call and wont be ciphered as described in point 7 above.

14- After step 52 the AuthParams will be send by Eutra cell A (as all the signalling is performed at CELL A).

Note: For these changes prose CR will be submitted at the enxt RAN5 meeting.

	Summary of change
	1- New template cdr_TAU_Request_92121d is added.
2- The Code is commented for the different PLMN.
3- The different variables are declared to be used in activating the default bearer.
v_PdnConnectivityRequest,v_EIT_Flag,v_EPS_TI,v_Pco,v_APN,v_EPS_ContextStatus

4- The power of EUTRA cell is made Serving to allow the UE to camp on the EUTRA cell.

5- v_PdnConnectivityRequest,v_EIT_Flag,v_EPS_TI,v_Pco,v_APN,v_EPS_ContextStatus is computed from the attach request to used in activating the default bearer.

6- The activate default bearer context request is sent with Attach accept (Piggybacking).

7- The Extended service request is made not to be ciphered.

8- At step 23 the EUTRA cell is made Non Suitable.

9- The Authparams is send by EUTRA cell A.

10- The EUTRA cell is made serving

11- The The Tracking area update procedure is used to contain the update type “tsc_EpsUpdate_Combined_IMSIAttach” and in the TAU accept sending the CSnotPreferred in Additional update.

12- The RRC connection is released after TAU procedure.

13- The Extended service request is made of MO call and not ciphered.

14- The AuthParams is send by Eutra cell A.

	Source of change
	NAS_Attach_UG.ttcn


Before change:

	  function f_TC_9_2_1_2_1d_EUTRA() runs on EUTRA_PTC

   {

    //Equip UE with a USIM containing default values except for those shown in Table 9.2.1.2.1d.3.1-1

    f_UT_USIM_Insert ( UT, "36.523-1 Table 9.2.1.2.1d.3.1-1" );

    // create cell and send out system information

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c4);

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_UT_ConfigureCombinedAttach (UT, DATA_CENTRIC);

    // Preamble: The UE is in state Switched OFF (state 1)

    // ==========================================================================

    f_EUTRA_Preamble_InterRAT(eutra_CellA, STATE2_IDLEUPDATE, UTRAN, Combined); // @sic R5s110176 sic@

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_CellA, RRC_IDLE, -, Combined); // @sic R5s110176 sic@

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_CellA, cs_HPLMN_002_11 ); //PLMN2 = MCC=002, MNC= 11

    f_EUTRA_ModifySysinfoUE_Off (eutra_CellA);
    // At T0 set cell type of Cell A to the "Serving cell" and Cell 5 to Suitable cell

    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

    f_EUTRA_TestBody_Set(true);

    fl_EUTRA_9_2_1_2_1d_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    // UE finishes on UTRAN

    f_EUTRA_ReleaseAllCells();

   }// function f_TC_9_2_1_2_1d_EUTRA

   /*

    * @desc      test body of TC_9_2_1_2_1d

    */

  function fl_EUTRA_9_2_1_2_1d_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti (eutra_CellA);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode (eutra_CellA);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(eutra_CellA);

    var template (omit) MobileIdentity v_MSId := f_GetMSId (Combined);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, Combined);

    var NAS_MSG_Indication_Type v_NasInd;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;

    //Synchronise with UTRAN PTC and get Cell Parameters for UTRAN cell

    v_UTRANSysInfo := f_WaitForIRATCoOrd_SysInfo (UTRAN);

    //@siclog "Step 1" siclog@

    // The UE is powered up or switched on.

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA); // @sic R5s120065 sic@

    // @siclog "Step 2-16" siclog@

    //Steps 2-16 of the registration procedure described in TS 36.508 table 4.5.2.3-1 are performed.

    v_NasInd := f_EUTRA_IdleUpdated_Step1_4 (eutra_CellA, NORMAL, Combined);

    f_EUTRA_IdleUpdated_Step5_13( eutra_CellA, STATE2_IDLEUPDATE, v_NasInd);
    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS (eutra_CellA,

                                                      tsc_RRC_TI_Def,

                                                      cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                                     cs_ATTACH_ACCEPT_Common (tsc_EAttachType_Combined,

                                                                                              cs_GprsTimer_v(tsc_GprsTimerUnit_deact, tsc_Spare5),

                                                                                              cs_TAIListNonConsecutive_lv (v_PLMN, {bit2oct(v_TAC)}),

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                                              v_LAI,

                                                                                              v_MSId,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              cs_EPS_NwkFtSup_IMSVoiceNotSupport,

                                                                                              cs_AdditionalUpdateResult_CSnotPrefer)));
    f_EUTRA_IdleUpdated_Step16_17 (eutra_CellA, true);

    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

    //@siclog "Step 18" siclog@

    //Transmit a Paging message to the UE on Cell 1 using S-TMSI with CN domain indicator set to "CS"

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    f_EUTRA_UE_Page (eutra_CellA,

                     cs_DomainNasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams.MME_Code,

                                                                              v_GutiParams.M_TMSI),

                                                  cs));

    //@siclog "Step 19a1" siclog@

    //IF the UE needs to request upper layer input before accepting the CS fallback, the incoming CS call is accepted at the UE through MMI or AT command.

    if ( not pc_ImmConnect ) {

      f_UT_AnswerCSCall (UT);

    }

    //@siclog "Step 20" siclog@

    //Receieve RRCConnectionRequest

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 21" siclog@

    //Send RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    //@siclog "Step 22" siclog@

    //receive RRCConnectionSetupComplete containing an EXTENDED SERVICE REQUEST with Service Type IE set to "mobile terminating CS fallback or 1xCS fallback"

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?, omit),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                 cr_508_EXT_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get(),

                                                                                            bit2oct(v_GutiParams.M_TMSI)))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22");

    //Set the Redirect Information in the RRCConnectionRelease message (step 1)

    if ( f_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD ) {// @sic R5-115770 sic@

      v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraFdd ( v_UTRANSysInfo.Utran[0].FDD.Freq );//[0] corresponds to Utran cell 5 according to the Init-function.

    }

    else {

      v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraTdd ( v_UTRANSysInfo.Utran[0].TDD.Freq );

    }

    //@siclog "Step 23" siclog@

    //Transmit RRC Connection Release message

    f_EUTRA_RRC_ConnectionRelease_Common ( eutra_CellA,

                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,

                                                                     other,

                                                                     v_RedirectedCarrierInfo,

                                                                     omit ) );

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);
    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

    //@siclog "Step 43-48" siclog@

    f_EUTRA_TrackingAreaUpdate(eutra_CellA);
    f_UT_RequestCSCall (UT);

    //@siclog "Step 49" siclog@

    //Receieve RRCConnectionRequest

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 50" siclog@

    //Send RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    //@siclog "Step 51" siclog@

    //receive RRCConnectionSetupComplete containing an EXTENDED SERVICE REQUEST with Service Type IE set to "mobile terminating CS fallback or 1xCS fallback"

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?, omit),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                                 cr_508_EXT_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get(),

                                                                                            bit2oct(v_GutiParams.M_TMSI)))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 51");

    //@siclog "Step 52" siclog@

    //Transmit RRC Connection Release message

    f_EUTRA_RRC_ConnectionRelease_Common ( eutra_CellA,

                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,

                                                                     other,

                                                                     v_RedirectedCarrierInfo,

                                                                     omit ) );

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);
    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

  }// function fl_tc_9_2_1_2_1d_Body




After change:

	template NAS_UL_Message_Type cdr_TAU_Request_92121d(EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                      NAS_KsiValue p_KsiValue,

                                                      template AdditionalUpdateType p_AdditionalUpdateType

                                                      )

    modifies cr_508_TAU_Request :=

  { /* @status    APPROVED */

    tRACKING_AREA_UPDATE_REQUEST := {

      oldPtmsiSignature          := cr_PTMSI_SignatureAny, // @sic R5-115634 sic@

      additionalGuti             := cr_MobileIdAny ('50'O) ifpresent,

      nonce                      := cr_NonceUE_Any

    }

  }; //Note1
function f_TC_9_2_1_2_1d_EUTRA() runs on EUTRA_PTC

   {

    //Equip UE with a USIM containing default values except for those shown in Table 9.2.1.2.1d.3.1-1

    f_UT_USIM_Insert ( UT, "36.523-1 Table 9.2.1.2.1d.3.1-1" );

    // create cell and send out system information

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c4);

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_UT_ConfigureCombinedAttach (UT, DATA_CENTRIC);

    // Preamble: The UE is in state Switched OFF (state 1)

    // ==========================================================================

    f_EUTRA_Preamble_InterRAT(eutra_CellA, STATE2_IDLEUPDATE, UTRAN, Combined); // @sic R5s110176 sic@

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_CellA, RRC_IDLE, -, Combined); // @sic R5s110176 sic@

    //f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_CellA, cs_HPLMN_002_11 ); //PLMN2 = MCC=002, MNC= 11

   // f_EUTRA_ModifySysinfoUE_Off (eutra_CellA); //Note2
    // At T0 set cell type of Cell A to the "Serving cell" and Cell 5 to Suitable cell

    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

    f_EUTRA_TestBody_Set(true);

    fl_EUTRA_9_2_1_2_1d_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    // UE finishes on UTRAN

    f_EUTRA_ReleaseAllCells();

   }// function f_TC_9_2_1_2_1d_EUTRA

   /*

    * @desc      test body of TC_9_2_1_2_1d

    */

  function fl_EUTRA_9_2_1_2_1d_Body() runs on EUTRA_PTC

  {

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti (eutra_CellA);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode (eutra_CellA);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(eutra_CellA);

    var template (omit) MobileIdentity v_MSId := f_GetMSId (Combined);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, Combined);

    var NAS_MSG_Indication_Type v_NasInd;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var boolean v_EIT_Flag := false;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var template (omit) AccessPointName v_APN := omit;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@
//Note3
    //Synchronise with UTRAN PTC and get Cell Parameters for UTRAN cell

    v_UTRANSysInfo := f_WaitForIRATCoOrd_SysInfo (UTRAN);

    f_EUTRA_SetCellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ); //Note4
    //@siclog "Step 1" siclog@

    // The UE is powered up or switched on.

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA); // @sic R5s120065 sic@

    // @siclog "Step 2-16" siclog@

    //Steps 2-16 of the registration procedure described in TS 36.508 table 4.5.2.3-1 are performed.

    v_NasInd := f_EUTRA_IdleUpdated_Step1_4 (eutra_CellA, NORMAL, Combined);

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    // Get the PCO to be used later (TS 36.508 v.8.2.2 Table 4.7.3-6)

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    v_APN := f_EUTRA_IdleUpdated_Step5_13( eutra_CellA, STATE2_IDLEUPDATE, v_NasInd); //Note5
    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS (eutra_CellA,

                                                      tsc_RRC_TI_Def,

                                                      cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                     cs_ATTACH_ACCEPT_Common (tsc_EAttachType_Combined,

                                                                                              cs_GprsTimer_v(tsc_GprsTimerUnit_deact, tsc_Spare5),

                                                                                              cs_TAIListNonConsecutive_lv (v_PLMN, {bit2oct(v_TAC)}),

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                                              v_LAI,

                                                                                              v_MSId,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              omit,

                                                                                              cs_EPS_NwkFtSup_IMSVoiceNotSupport,

                                                                                              cs_AdditionalUpdateResult_CSnotPrefer),

                                                                    cs_508_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                                               v_EPS_TI,

                                                                                                                               v_APN,

                                                                                                                               f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas (v_Pco)),

                                                                                                                               -,

                                                                                                                               f_GetDefPCOforPCSCFAddress (v_Pco)))); //Note6
    f_EUTRA_IdleUpdated_Step16_17 (eutra_CellA, true);

    f_SendIRATCoOrd (UTRAN, cms_IRAT_Trigger);

    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

    //@siclog "Step 18" siclog@

    //Transmit a Paging message to the UE on Cell 1 using S-TMSI with CN domain indicator set to "CS"

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    f_EUTRA_UE_Page (eutra_CellA,

                     cs_DomainNasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParams.MME_Code,

                                                                              v_GutiParams.M_TMSI),

                                                  cs));

    //@siclog "Step 19a1" siclog@

    //IF the UE needs to request upper layer input before accepting the CS fallback, the incoming CS call is accepted at the UE through MMI or AT command.

    if ( not pc_ImmConnect ) {

      f_UT_AnswerCSCall (UT);

    }

    //@siclog "Step 20" siclog@

    //Receieve RRCConnectionRequest

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 21" siclog@

    //Send RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    //@siclog "Step 22" siclog@

    //receive RRCConnectionSetupComplete containing an EXTENDED SERVICE REQUEST with Service Type IE set to "mobile terminating CS fallback or 1xCS fallback"

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?, omit),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected, //Note7
                                                                 cr_508_EXT_SERVICE_REQUEST(f_EUTRA_SecurityKSIasme_Get(),

                                                                                            bit2oct(v_GutiParams.M_TMSI)))));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22");

    //Set the Redirect Information in the RRCConnectionRelease message (step 1)

    if ( f_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD ) {// @sic R5-115770 sic@

      v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraFdd ( v_UTRANSysInfo.Utran[0].FDD.Freq );//[0] corresponds to Utran cell 5 according to the Init-function.

    }

    else {

      v_RedirectedCarrierInfo := cs_RedirectionInfo_UtraTdd ( v_UTRANSysInfo.Utran[0].TDD.Freq );

    }

    //@siclog "Step 23" siclog@

    //Transmit RRC Connection Release message

    f_EUTRA_RRC_ConnectionRelease_Common ( eutra_CellA,

                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,

                                                                     other,

                                                                     v_RedirectedCarrierInfo,

                                                                     omit ) );

   f_EUTRA_SetCellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ); //Note8
    f_EUTRA_SendAuthParameters(eutra_CellA, UTRAN); //Note9
    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

    f_EUTRA_SetCellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ); //Note10
    //@siclog "Step 43-48" siclog@

    //f_EUTRA_TrackingAreaUpdate(eutra_CellA);

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas ( eutra_CellA,

                                                 tsc_RRC_TI_Def,

                                                 ?,   //@sic R5s100539 sic@

                                                 cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                   cdr_TAU_Request_92121d (tsc_EpsUpdate_Combined_IMSIAttach,

                                                                                       f_EUTRA_SecurityKSIasme_Get ( ),

                                                                                       f_GetAdditionalUpdateType (Combined)))); // @sic R5-103681 sic@

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(eutra_CellA,

                                               v_GutiParams,

                                               tsc_EpsUpdate_Combined_TaLaUpdate,

                                               v_EPS_ContextStatus,

                                               cs_AdditionalUpdateResult_CSnotPrefer, // @sic R5s110078 sic@

                                               Combined,

                                               omit); //Note11
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA); //Note12                                           

    f_UT_RequestCSCall (UT);

    //@siclog "Step 49" siclog@

    //Receieve RRCConnectionRequest

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    //@siclog "Step 50" siclog@

    //Send RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    //@siclog "Step 51" siclog@

    //receive RRCConnectionSetupComplete containing an EXTENDED SERVICE REQUEST with Service Type IE set to "mobile terminating CS fallback or 1xCS fallback"

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?, omit),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cr_EXT_SERVICE_REQUEST_MO(f_EUTRA_SecurityKSIasme_Get(),

                                                                                            bit2oct(v_GutiParams.M_TMSI))))); //Note13
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 51");

    //@siclog "Step 52" siclog@

    //Transmit RRC Connection Release message

    f_EUTRA_RRC_ConnectionRelease_Common ( eutra_CellA,

                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,

                                                                     other,

                                                                     v_RedirectedCarrierInfo,

                                                                     omit ) );

    f_EUTRA_SendAuthParameters(eutra_CellA, UTRAN); //Note14
    f_WaitForIRATCoOrd_Trigger(UTRAN); // wait for trigger from UTRAN side

  }// function fl_tc_9_2_1_2_1d_Body


4.3 Change 2

	Testcase name
	9.2.1.2.1d

	Reason for change
	1- At the initial stage the Default PLMN can be used for this Test Case.

2- The wait timer will be made 90 seconds as per the standard for checking the RRC connection request at UTRA side.

3- The different variables will be used to capture the RAU after MT call and to activate security.

4-  In the initially the cell power of UTRAN will be made not suitable to not allow the UE to camp on UTRAN cell.

5- Then after the step 17 UTRA cell will be made suitable to check that UE is not trying to camp on UTRA cell.

6- At step 24-42 the MT call will be disconnected by using the the MT TI value.

3GPP TS 24.007 11.2.3.1.3 describes the transaction identifier (TI). During the disconnect of MT call, the constant cs_TI_MT should be used in f_UTRAN_CC_Disconnect().

7- After step 42 the RRC connection will be released and then established to allow the UE to perform the RAU procedure. .(As per TS 36.523-3 table 10.1.3.1-1)

8- After step 42 The UE will perform the RAU procedure as it entered UTRAN from EUTRA with the activation of security.(As per TS 36.523-3 table 10.1.3.1-1)

Note: A prose CR will be raised for all these changes at the next RAN5 meeting.

	Summary of change
	1- At the initial stage the Default PLMN is used for this Test Case.

2- The wait timer is made 90 seconds as per the standard for checking the RRC connection request at UTRA side.

3- The different variables are used to capture the RAU after MT call and to activate security.

v_START_Value & v_RRC_DataInd

4-  In the initially the cell power of UTRAN is made not suitable to not allow the UE to camp on UTRAN cell.

5- Then after the step 17 UTRA cell is made suitable to check that UE is not trying to camp on UTRA cell.

6- At step 24-42 the MT call is disconnected by using the the MT TI value.

7- After step 42 the RRC connection is released and then established to allow the UE to perform the RAU procedure.

8- After step 42 the RAU procedure is used as it entered UTRAN from EUTRA with the activation of security.

	Source of change
	NAS_Attach_UG_UTRA.ttcn


Before change:

	  function f_TC_9_2_1_2_1d_UTRAN () runs on UTRAN_PTC

  { // @desc Combined attach procedure / Success / EPS and CS Fallback not preferred/ data centric UE

    var PLMN_Identity v_PLMN2 := { mcc := {0,0,2},  mnc :={1,1} };//PLMN2

    timer t_WaitS := 20.0;
    f_UTRAN_Init(EUTRA_UTRAN); // This is still only a 2 PTC test

    f_UTRAN_CellInfo_SetPLMNIdentity(utran_Cell5, v_PLMN2 ); //PLMN2

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationSuitableNeighbourCell); //T0
    f_UTRAN_TestBody_Set( true );

    //@siclog "Step 17" siclog@

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    //+   Check: Does the UE try to access the UTRAN Cell 5 in the next 20s?

    t_WaitS.start;

    alt { // if the UE tries to access the cell, it will send an RRC Conn Req

      [] U_TM.receive(car_RRC_ConnReq(utran_Cell5,

                                      tsc_RB0,

                                      cr_RRC_RrcConnReqAny))

        {

          t_WaitS.stop;

          f_UTRAN_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 17");

        }

       [] t_WaitS.timeout {}

    }

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 24-42" siclog@

    // Steps 1-19 of the procedure in table 6.4.3.7.1-1 of 36.508

    f_UTRAN_CS_Fallback_WithRedirection (utran_Cell5);

    f_UTRAN_CC_Disconnect (utran_Cell5);
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 53-71" siclog@

    // Steps 1-19 of the procedure in table 6.4.3.7.1-1 of 36.508

    f_UTRAN_CS_Fallback_WithRedirection (utran_Cell5, false);

    // Test case now finished on UTRAN side

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_TestBody_Set( false );

    f_UTRAN_Postamble (utran_Cell5, U5_CSCALL);

    } // f_TC_9_2_1_2_1d_UTRAN


After change:

	  function f_TC_9_2_1_2_1d_UTRAN () runs on UTRAN_PTC

  { // @desc Combined attach procedure / Success / EPS and CS Fallback not preferred/ data centric UE

   // var PLMN_Identity v_PLMN2 := { mcc := {0,0,2},  mnc :={1,1} };//PLMN2 //Note 1
    timer t_WaitS := 90.0; //Note 2
    var START_Value v_START_Value := '00000000000000000000'B;

    var RRC_DATA_IND v_RRC_DataInd; //Note 3
    f_UTRAN_Init(EUTRA_UTRAN); // This is still only a 2 PTC test

    //f_UTRAN_CellInfo_SetPLMNIdentity(utran_Cell5, v_PLMN2 ); //PLMN2//Note 1
    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationNonSuitableCell); //T0//Note 4
    f_UTRAN_TestBody_Set( true );

    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);

    //@siclog "Step 17" siclog@

    f_WaitForIRATCoOrd_Trigger (EUTRA);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationSuitableNeighbourCell); //T0//Note 5
    //+   Check: Does the UE try to access the UTRAN Cell 5 in the next 20s?

    t_WaitS.start;

    alt { // if the UE tries to access the cell, it will send an RRC Conn Req

      [] U_TM.receive(car_RRC_ConnReq(utran_Cell5,

                                      tsc_RB0,

                                      cr_RRC_RrcConnReqAny))

        {

          t_WaitS.stop;

          f_UTRAN_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 17");

        }

       [] t_WaitS.timeout {}

    }

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 24-42" siclog@

    // Steps 1-19 of the procedure in table 6.4.3.7.1-1 of 36.508

    f_UTRAN_CS_Fallback_WithRedirection (utran_Cell5);

    f_UTRAN_CC_Disconnect (utran_Cell5, cs_TI_MT);

    f_UTRAN_RRC_ConnRel( utran_Cell5, cell_Dch); //Note 6
    f_UTRAN_RRC_ConnEst ( utran_Cell5);

   alt { 

   [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                   tsc_RB3,

                                                   cr_U_RA_UpdReqAny (cr_GMM_UpdateType(?, tsc_I_UpdateTypeCombinedWithIMSI),

                                                                      cr_RAI (omit),

                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                      ? ))) // @sic R5s110216 sic@

              -> value v_RRC_DataInd

       { 

        v_START_Value := v_RRC_DataInd.ttcn_start;

        f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value);

        f_UTRAN_GMM_RAU_Without_RAUReq ( utran_Cell5, v_START_Value, tsc_I_UpdateTypeCombined, false, false );

        f_UTRAN_RRC_ConnRel( utran_Cell5, cell_Dch);

        }//Note 7
        }
    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 53-71" siclog@

    // Steps 1-19 of the procedure in table 6.4.3.7.1-1 of 36.508

    f_UTRAN_CS_Fallback_WithRedirection (utran_Cell5, false);

    // Test case now finished on UTRAN side

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_TestBody_Set( false );

    f_UTRAN_Postamble (utran_Cell5, U5_CSCALL);

    } // f_TC_9_2_1_2_1d_UTRAN


5 Execution Log Files

5.1 Nvidia Icera 410

The Nvidia Icera 410 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_1_2_1d_ICERA.htm:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

5.2 Qualcomm MDM 9200

The Qualcomm UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_1_2_1d_Qualcomm.htm:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6. References

	[1]
	R5s120157: This archive comprises html format execution log file.

